LESSON 1
THE PIXEL AND ITS COLOR REGISTER
Version 1.0

There are 4 screen display parameters that must be initialized when the Bally/Astrocade console is powered on. The motherboard ROM executes a power up routine to initialize these screen parameters prior to displaying the system menu. The parameters are changed or initialized in all ROM cartridges and can be changed any time during execution of a program.

The 4 parameters are the color registers, the horizontal color boundary, the border background color and vertical blanking, all of which will be described in the first 3 lessons of this series.

A pixel (picture element) is a tiny square displayed on a TV or monitor screen. In the case of the Bally/Astrocade home computer system, which utilizes 1970s technology, a pixel can be displayed in one of only 4 possible colors. This technology defines a pixel with 2 bits of screen RAM. So, one byte of screen RAM can define 4 pixels. The Nutting Manual's system description describes and illustrates the screen map on pages 85-86 for the Bally/Astrocade home computer. The screen's resolution is a mere 160 x 102 pixels. The illustration on page 86 breaks down the byte at 4000H, designating the right most pixel as pixel 0 and the left most pixel as pixel 3.

The two bits for a pixel are mapped to a specific 8 bit color register. Since the color register is 8 bits long, it has a range of 256 possible colors with values from 00-FFH. The color register defines the color in bits 3 thru 7 and the color intensity in bits 0 thru 2. Intensity 000 is darkest and 111 is lightest.

A 256 color palette is posted on the Bally Alley.com website in the Machine Language/Astrocade ML docs section.

https://ballyalley.com/ml/ml_docs/astrocade_palette.html

The palette values in the first vertical column are 00,08,10,18,20,38H and so on. The value increments for each additional column to the right.

Eight color registers are available and programmable. This Lesson 1 will introduce only the four color registers 7,6,5 and 4. The remaining 4 color registers will be presented in Lesson 2. Each color register can be individually or collectively programmed using a Z80 output instruction. The following list indicates the pixel to color register map for registers 7 thru 4.

2 bit		Mapped to	Set Up via
Pixel		Color		Output port
		Register	Hex Number

11		    7		   07
10		    6		   06
01		    5		   05
00		    4		   04

Note that these color registers are internal to a custom hardware data chip. They are NOT registers internal to the Z80 CPU. The custom hardware is set up to receive a color set up request using a Z80 output instruction.

The MLM pixel color table (list) is located at 26B9H. This color table is documented in the MLM users manual, Appendix A: MLM Source Listing, page 0022.

The MLM pixels are defined as:

2 bit		Mapped to	Hex	Colr		Used for
Pixel		Color Reg	Value

11		    7		A3	Green
10		    6		52	Red
01		    5		07	White		Text
00		    4		00	Black		Background



EXAMPLE 1-1


Program Description

Change the MLM background color to light blue and the text color to black.

+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++ Example Program 1-1

4E10H  00 00 00     NOP         3 NOPs. NOP = No Operation by Z80.
                                Leave room for Example Program 1-2.

4E13H  3E F3        LD A,F3H    A = color value = light blue = F3H

4E15H  D3 04        OUT (04),A  Output this color to clr reg 4 (background)

4E17H  3E 00        LD A,00H    A = color value = black = 00H

4E19H  D3 05        OUT (05),A  Output this color to clr reg 5 (text)

4E1BH  C9           RET         Return to MLM

+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

The Example Program 1-1 has a total of 12 ML hex byte codes in it. The ML program begins at screen address 4E10H and ends at 4E1BH. Use MLM to write (load) these 12 bytes into screen RAM beginning at 4E10H. When you are finished loading this program, use MLM and list the program to check for digital errors in the program bytes. If there are any errors, use MLM's write command again to change (correct) the byte at the address where the error is present. To correct the byte, point to the address of that byte and write the correct hex code in that address. When all bytes in this program are correct, use MLM to call (execute) this program. The background color should change to light blue and the text color to black. Congratulations! You just ran a program in machine language.


Beginner Exercise

You can also use the color palette on the Bally Alley and choose some other colors to change the MLM text or background color. Select a color on the palette table and read its hex color value. Use the MLM write and read commands to change the hex color value in the program and to confirm the byte change. You can also use the MLM list command instead of the read command.

The hex color value for the background is at 4E14H.
The hex color value for the text is at 4E18H.

Note the color on the Bally Alley palette table may not quite match exactly what you see on your screen.



EXAMPLE 1-2


Program Description

Clear the screen.
Change the MLM background and text colors back to black and white.
Write a byte on the top screen line, about midway on that line, so you can experiment with (change) the pixel colors in that byte.


Beginner's Comments

In this example, the screen address of the byte to write will be manually determined. In the next EXAMPLE 1-3, ML will be used to compute that screen address.

Reference page 86 of the Nutting Manual's system description, showing the low resolution screen map. One screen line is 40 bytes wide. So, half of the line is 20 bytes. Since the top screen line begins with address 4000H, we need to add 19 more bytes to give us the ending address (the 20th byte) for half a line. Changing the decimal number 19 to hex, give us 13H. So, the ending screen address for half of the top line is 4013H. Note that the ACTUAL midpoint of the screen for the top line is BETWEEN address 4013H and 4014H as shown below.

Left side of line (top line) | Right side of line (top line)
20 bytes			     | 20 bytes
4000H			     4013H | 4014H			 4027H

In this example, we could use 4013H or 4014H which are about midway on this top line. Let's use 4013H in the program below.

+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
Example Program 1-2

CLEAR SCREEN

4E10H CD 47 23	CALL CLEAR		Use MLM CLEAR screen sub at 2347H
						Swap address in program with LO byte
						47H first
SET MLM BACKGROUND AND TEXT COLORS
BLACK AND WHITE
SIMILAR TO EXAMPLE PROGRAM 1-1

4E13H	3E 00		LD A,00		A = Bckgnd color value = black = 00H
	D3 04		OUT (04),A		Output it to Clr Reg 04
4E17H	3E 07		LD A,07		A = Text clr value = white = 07H
	D3 05		OUT (05),A		Output it to Clr Reg 05

WRITE A BYTE MIDWAY ON TOP LINE OF SCREEN RAM

4E1BH 21 13 40	LD HL,4013H		Point HL at screen address
						Use swapped format in program
4E1EH	3E 82		LD A,PBYTE		A = pixel byte to write = 82H
						  = 10  00  00  10
						    red blk blk red
4E20H	77		LD (HL),A		Write the byte to screen RAM
4E21H	C9		RET			Return to MLM

+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++


Beginner Exercises

Now would be a good time to practice some more programming in ML. You can experiment and program the pixels in screen address 4013H with variations in pixel colors. Simply change the value of the "pixel byte to write" (PBYTE) at the program address 4E1FH (byte 82H in program). See the listing below for  some sample pixel color variations.

Pixel Byte		Pixel Colors
To Write		To Display
(Hex Value)

    C3	   	11  00  00  11
			Grn Blk Blk Grn

    42 		01  00  00  10
			Wht Blk Blk Red

    BB		10  11  10  11
			Red Grn Red Grn

    AA		10  10  10  10
			Red Red Red Red

    FF		11  11  11  11
			Grn Grn Grn Grn


You can also select different colors from the palette color table and map them to the color registers 7,6,5 or 4 by altering the load and respective output instruction beginning at the program's address 4E13 or 4E17H.

Instead of writing a byte to screen address 4013H, write a pixel byte to the corner screen address 4000H or 4027H. Try writing a pixel byte on the 2nd screen line which begins at address 4028H. Don't forget to load the low order (LO) byte in RAM first, then the high order (H0) byte. Refer to the above program address at 4E1BH.

One last experiment you can run is to write not just 1 byte, but write 2 or 3 bytes to the 1st (top) or 2nd screen lines at, for example, addresses 4000, 4013, 4027 or 4028H. Use the above program as a guide to add the necessary load instructions to the program to run your experiment. These additional load instructions would be similar to the load instruction at 4E1BH and 4E20H. For example, use

LD HL,4027H
LD (HL),A

Place these additional load instructions just before the return (RET) instruction.




EXAMPLE 1-3


Program Description

Same as Example Program 1-2, except use ML to compute the "byte to write" screen address.
Place the MLM Breakpoint Program command (CFH) within the program to confirm the computation sum in HL is 4013H. See the MLM users manual, page 4-3 for details on MLM's Breakpoint Program.

+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++
Example Program 1-3

Use Example Program 1-2 from 4E10 to 4E1AH as listed below.

4E10H CD 47 23
4E13H 3E 00
      D3 04
4E17H 3E 07
4E19H D3 05

COMPUTE "BYTE TO WRITE" SCREEN ADDRESS (MIDWAY ON TOP LINE)

4E1BH 21 00 40 	LD HL,4000H		Point HL at 4000H
      01 13 00	LD BC,0013H		BC = 19 bytes
						Use swapped format
4E21H 09		ADD HL,BC		Compute end address at 1/2 line
						Add 19 bytes to HL = 4000H
						Sum in HL = 4013H

WRITE THE PIXEL BYTE

4E22H	3E 82		LD A,PBYTE		A = pixel byte = 10 00 00 10 = 82H
4E24H	77		LD (HL),A		Write the byte to screen RAM
4E25H	C9		RET			Return to MLM

+++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++++

Place (Write) the MLM Breakpoint Program command byte CFH at an address after the ADD HL,BC instruction, at 4E22H, 4E24H or 4E25H. Then run the Example Program 1-3 again. You should see a Breakpoint report displayed as indicated below if you place the command byte at 4E25H.

BRPT ADDR:4E25
A:82 BC:0013 DE=ignore HL=4013 ADDR:4E10

Note the Breakpoint report shows the computation sum in HL as address 4013H


Beginner Exercises

Remove the Breakpoint command byte that you placed in the program, i.e., restore the program as listed above. Now place the Breakpoint command at address 4E21H and see if you can figure out why the Breakpoint command is reporting the hex values in the registers A, BC and HL displayed on the screen.

Important! When you place the Breakpoint command byte CFH within a program, to debug a program, you must pay attention to the address you are placing the command byte in. Some Z80 instructions have more than 1 byte of coding. Place the Breakpoint command byte only at the address of the FIRST byte of coding. If you place CFH in the other coding bytes of the Z80 instruction, the Z80 will not execute the Breakpoint program and will read the program instruction that you altered as written and will likely crash the program. This hex code CFH is actually a Z80 restart instruction. Z80 restart instructions will be presented in a later lesson. Try this. Restore the program again and this time place the Breakpoint command byte CFH at address 4E1FH, run the program and watch the "byte to write" written to another screen RAM address other than 4013H. Can you figure out why? The program will, in this case, write the pixel byte at address 40CFH. Why? Because the Z80 now reads the original Mneumonic instruction
LD BC,0013H as LD BC,00CFH, which is the new instruction you just altered. So, the program, in this case, writes the pixel byte at screen address 40CFH.

